### ----------------------------------------------------
### -- GIO
### -- Licenses: GPL-3.0
### -- Authors: clifton.wood@gmail.com
### -- File: dev-t/02-buffered-input-stream.t
### ----------------------------------------------------
use v6.c;

use Test;

use GIO::Raw::Types;

use GLib::MainContext;

use GIO::BufferedInputStream;
use GIO::InputStream;
use GIO::MemoryInputStream;
use GIO::Task;

sub tests-init ($data = 'abcdefghijk', $size = 0, $enc = 'ASCII') {
  my $base    = GIO::MemoryInputStream.new-from-data( $data.encode($enc) );
  my $in      = $size ??  GIO::BufferedInputStream.new-sized($base, $size) !!
                          GIO::BufferedInputStream.new($base);


  ($data, $base, $in);
}

sub test-peek {
  my ($data, $base, $in) = tests-init;

  $in.fill(5);

  is  $in.available, 5,
      '5 Bytes read into buffer after call to .fill(5)';

  $in.fill(-1);

  is  $in.available, $data.chars,
      'All of $data available in buffer after call to .fill(-1)';

  {
    my $buf  = Buf.allocate(64, 0);
    my $peek = $in.peek($buf, 2, 3);

    is  $peek, 3,
        "Number of bytes peek'ed is 3";

    # Remove all NULs before comparison!
    # cw: This is a flapping failures. Find out why.
    is  $buf.decode.substr(0, $peek), 'cde',
        'Result of peek has "cde" in $buf';
  }

  {
    my $buf  = Buf.allocate(64, 0);
    my $peek = $in.peek($buf, 9, 5);

    is  $peek, 2,
        "Number of bytes peek'ed is 2";

    # cw: This is a flapping failures. Find out why.
    is  $buf.decode.substr(0, $peek), 'jk',
        'Result of peek has "jk" in $buf';
  }

  {
    my $buf  = Buf.allocate(64);
    my $peek = $in.peek($buf, 75, 3);

    is  $peek, 0,
        "Number of bytes peek'ed is 0 (outside range of \$data)";
  }
}

sub test-peek-buffer {
  my ($data, $base, $in) = tests-init;

  my $nf      = $in.fill($data.chars);
  my ($b, $c) = $in.peek-buffer(:all);

  is  $nf, $c,
      "Number of filled bytes equals number of bytes peek'd";

  #diag $b.perl;

  is  $b.decode('ASCII'), $data,
      'Buffer returned from peek is the same as original data';
}

sub test-set-buffer-size {
  my ($data, $base, $in) = tests-init;

  is  $in.buffer-size, 4096,
      'Default BufferedInputStream buffer-size is 4096';

  $in.buffer-size = 64;

  is  $in.buffer-size, 64,
      'After resetting, current buffer-size is 64';

  $in.fill($data.chars);
  my ($b, $c) = $in.peek-buffer(:all);
  $in.buffer-size = 2;
  is  $in.buffer-size, $c,
      "Peek'd buffer size is the same before buffer-size reset to 2";

  $in.unref;
  $in = GIO::BufferedInputStream.new-sized($base, 64);
  is  $in.buffer-size, 64,
      'Newly allocated BufferedInputStream has the proper buffer size of 64.';
}

sub test-read-byte {
  use GIO::Raw::Quarks;

  my ($data, $base, $in) = tests-init('abcdefgh');

  sub checkByte ($b) {
    if $b ~~ Str {
      is  $in.read-byte, $b.ord,
          "Return value from byte read is '{$b}'";
    } else {
      is  $in.read-byte, $b,
          "Return value from byte read is {$b}";
    }
    nok $ERROR,
        'No error detected after reading byte';
  }

  checkByte($_) for <a b c>;

  $in.skip(3);
  nok $ERROR,
      'No error detected after skipping 3 bytes';

  checkByte($_) for <g h>;
  checkByte(-1);

  $in.close;
  nok $ERROR,
      'No error detected after stream close';
  is  $in.read-byte, -1,
      'Return value from byte read is -1';

  ok  [&&](
        $ERROR,

        # The proper value for $G_IO_ERROR is 51. See c-helpers/time_t.
        # However, the domain returned in the error is a stable 67.
        # This goes AGAINST this line from the reference implementation!
        # https://github.com/GNOME/glib/blob/master/gio/tests/buffered-input-stream.c#L155
        # So, for now, we do not perform the domain check, as this works EVERYWHERE ELSE!
        # -Cliff

        #$ERROR.domain == $G_IO_ERROR,
        $ERROR.code   == G_IO_ERROR_CLOSED
      ),
      'Proper error was returned after byte read on closed stream';
}

sub test-read {
  my ($data, $base, $in) = tests-init(
    'abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVWXYZ',
    8,
    'ISO-8859-1'
  );

  is  $in.available, 0,
      'No available bytes in input stream';

  is  $in.fill(8), 8,
      'Number of bytes (8) requested for fill and number returned are the same';

  nok $ERROR,
      'No error resulted from call to .fill';

  is  $in.available, 8,
      'Input stream reports that 8 bytes are available';

  sub getChunk ($cmp = '') {
    my $buf  = Buf.allocate(20, 0);
    my $size = $cmp.chars;

    is  $in.read($buf, 16), $size,
        ".read operation returned the proper number of bytes ({$size})";

    is  $buf.decode.substr(0, $size), $cmp,
        ".read operation returned the proper value ('{$cmp}')" if $cmp;

    nok $ERROR,
        '.read operation did not incur an error';
  }

  getChunk('abcdefghijklmnop');

  is  $in.available, 0,
      'Input stream reports no available bytes after read.';

  getChunk('qrstuvwxyzABCDEF');
  getChunk('GHIJKLMNOPQRSTUV');
  getChunk('WXYZ');
  getChunk;
}

my $result;
sub result-cb ($, $r, $) {
  CATCH { default { .message.say; .backtrace.summary.say } }

  $result = GIO::Task.new($r).ref;
}

# cw: $buffer performs best outside the scope of test-read-async()
my $buffer;

sub test-read-async {
  my ($data, $base, $in) = tests-init(
    'abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVWXYZ',
    8,
    'ISO-8859-1'
  );

  is  $in.available, 0,
      'No bytes available after stream init.';

  sub getChunk ($cmp = '', $size = 16) {
    if $size == 8 {
      $in.fill-async($size, G_PRIORITY_DEFAULT, &result-cb);
    } else {
      $buffer = Buf.allocate(20, 0);
      $in.read-async($buffer, $size, G_PRIORITY_DEFAULT, &result-cb);
    }

    repeat { GLib::MainContext.iteration } until $result;

    is  $in.fill-finish($result), $size,
        "Async read operation returned proper number of bytes ({$size})";

    is  $buffer.decode('ISO-8859-1').substr(0, $cmp.chars), $cmp,
        "Async read operation returned proper data ('{$cmp}')" if $cmp;

    nok $ERROR,
        'Async operation did not incur an error';

    $result = Nil;
  }

  getChunk('', 8);
  getChunk($_, .chars) for $data.comb(16);
  getChunk('', 0);
}

sub test-skip {
  my ($data, $base, $in) = tests-init(
    'abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVXYZ',
    5,
    'ISO-8859-1'
  );

  for <a b c> {
    is  $in.read-byte, $_.ord,  "Byte read in is '$_'";
    nok $ERROR,                 'No read error occurred.';
  }

  for <7 k 10 v 20 Q>.rotor(2) -> ($s, $l) {
    is  $in.skip($s), $s,       "Skip operation returned {$s} bytes";
    nok $ERROR,                 'No read error occurred.';
    is  $in.read-byte, $l.ord,  "Next byte read was a '{$l}'";
    nok $ERROR,                 'No read error occurred.';
  }

  for 8, 0 {
    is  $in.skip(10), $_,
        "After attempted skip of 10 bytes, only {$_} were really skipped.";

    nok $ERROR, 'No read error occurred.';
  }
}

sub test-skip-async {
  my ($data, $base, $in) = tests-init(
    'abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVXYZ',
    5,
    'ISO-8859-1'
  );

  for <a b c> {
    is  $in.read-byte, $_.ord,  "Byte read in is '$_'";
    nok $ERROR,                 'No read error occurred.';
  }

  for <7 k 10 v 20 Q>.rotor(2) -> ($s, $l) {
    $in.skip-async($s, &result-cb);

    repeat { GLib::MainContext.iteration } until $result;

    is  $in.skip-finish($result), $s, "Async Skip operation returned {$s} bytes";
    nok $ERROR,                       'No read error occurred.';
    is  $in.read-byte, $l.ord,        "Next byte read was a '{$l}'";
    nok $ERROR,                       'No read error occurred.';

    $result = Nil;
  }

  for 8, 0 {
    $in.skip-async(10, &result-cb);

    repeat { GLib::MainContext.iteration } until $result;

    is  $in.skip-finish($result), $_,
        "Attempted skip of 10 bytes returned {$_} bytes";

    nok $ERROR,
        'No read error occurred.';

    $result = Nil;
  }
}

sub test-seek {
  my ($data, $base, $in) = tests-init(
    'abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVXYZ',
    4,
    'ISO-8859-1'
  );

  my $curPos = 0;
  sub seekTest($c, $o = 0, $t = G_SEEK_CUR) {
    if $o {
      my $rc = $in.seek($o, $t);

      given $t {
        when G_SEEK_CUR { $curPos += $o               }
        when G_SEEK_SET { $curPos  = $o               }
        when G_SEEK_END { $curPos  = $data.chars + $o }
      }

      nok $ERROR,             'No error received from seek operation';
      ok  $rc,                'Seek operation return code was non-zero';
      is  $in.tell, $curPos,  "Current position of stream is byte {$curPos}";
    }

    my $b = $in.read-byte;
    $curPos++;

    nok $ERROR,               'No error received from read operation';
    is  $b, $c.ord,           "Byte received from read is '{$c}'";
    is  $in.tell, $curPos,    "Current position of stream is byte {$curPos}";
  }

  seekTest('a');
  seekTest('c',  1);
  seekTest('b', -2);
  seekTest('i',  6);
  seekTest('d', -6);
  seekTest('i',  8, G_SEEK_SET);
  seekTest('Z', -1, G_SEEK_END);
}

sub test-close {
  {
    my ($data, $base, $in) = tests-init(
      'abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVXYZ',
      0,
      'ISO-8859-1'
    );

    ok  $in.close-base-stream, '.close-base-stream is enabled.';
    ok  $in.close,             '.close returns TRUE';
    nok $ERROR,                'No error was raised by .close';
    ok  $base.is-closed,       'Base Stream is closed';
  }

  {
    my ($data, $base, $in) = tests-init(
      'abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVXYZ',
      0,
      'ISO-8859-1'
    );

    $in.close-base-stream = False;

    ok  $in.close,             '.close returns TRUE';
    nok $ERROR,                'No error was raised by .close';
    nok $base.is-closed,       'Base Stream is NOT closed';
  }
}

plan 155;

test-peek;
test-peek-buffer;
test-set-buffer-size;
test-read-byte;
test-read;
test-read-async;
test-skip;
test-skip-async;
test-seek;
test-close;


